


Z80A-ECB/E1

Einkanal-Analogeingabe-Baugruppe

Version: 1.2

Datum: September/1984



Z80A-ECB/E1

Einkanal Analogeingabe-Baugruppe


A N W E N D E R H A N D B U C H

Baugruppe: 380

Version: 1.2


Stand: September/1984

Copyright by KONTRON MIKROCOMPUTER GmbH, Eching/München



I n h a l t:


	Seite
1. Beschreibung	1
2. Adressierung	2
2.1 Zuordnung der Steueradresse	2
3. Betriebsarten	3
3.1 Mögliche Benutzung der Baugruppe als Ein- kanalmeßsystem	3
3.1.1 Einkanalige Messungen, die unmittel- bar aufeinanderfolgen	3
3.1.2 Einkanalige Messungen mit beliebig wählbarem Meßintervall	4
3.2 Mögliche Benutzung der Baugruppe ECB/E1 als Mehrkanalmeßsystem	4
3.2.1 Mehrkanalige Messung mit unmittelbar aufeinanderfolgende Meßzyklen	4
3.2.2 Mehrkanalige Messung mit beliebig wählbarem Meßintervall	5
3.3 Beispielprogramme	6
4. Abgleichvorschrift	6
5. Steckerbelegung	7
6. Anschlußbeispiele	8



	Seite
7. Beschreibung der Treibermodule zur Analogeingabe-Baugruppe ECB/E1	9
7.1 Initialisierung der Analogeingabe-Baugruppe ECB/E1 und Programmierung des Spannungsreiches	9
7.2 Starten der Messung	10
7.3 Starten der Messung mit der Relaismulti-plexer-Baugruppe ECBS/R16	11
7.4 Überprüfung auf Ende der Messung; Warten auf Messungsende	12
7.5 Schema für die Parameterübergabe	13
8. Technische Daten	13
9. Besonderheiten der Baugruppe	14
10. Beispielprogramme	15
11. Treiberprogramm zur ECB/E1	20

Dieses Anwenderhandbuch ist mit größter Sorgfalt erstellt worden. Es wird jedoch keine Gewähr für die Freiheit von Fehlern und Irrtümern gegeben. Technische Änderungen bleiben vorbehalten.

Für alle Anfragen stehen Ihnen unsere Technischen Büros und Ihr Distributor zur Verfügung.



1. Beschreibung

Die Analogeingabe-Baugruppe ECB/E1 ist vorwiegend gedacht für das hochgenaue Messen von langsam sich veränderten Analogsignalen bei gleichzeitiger galvanischer Trennung des Prozesses von der Rechnerseite.

Damit bei vertretbarem Aufwand die hohe Auflösung und Genauigkeit realisiert werden kann, wird ein integriertes A/D-Wandlungsverfahren (Dual-Slope) angewandt.

Die Baugruppe erlaubt hochauflösende (16 Bit incl. VZ) isolierte Messungen analoger Größen in einem weiten programmierbaren Eingangsspannungsbereich, wählbar in 13 Stufen von ± 10 Millivolt bis ± 50 Volt. Die ECB/E1 ist einkanalig ausgeführt.

Die Eingangskanalkapazität kann durch vorgeschaltete Relais-Multiplexerkarten bis auf 128 Kanäle erweitert werden. Kernstück dieser Baugruppe ist ein Analogmodul, welches einen Dual-Slope A/D-Wandler, eine durch Software programmierbare Vorverstärkerstufe, eine isolierte Spannungsversorgung sowie eine isolierte Interfacelogik enthält.


Über diese Interfacelogik erhält das Modul von der mit LSI- und PAL-Bausteinen realisierten Steuerungseinheit alle notwendigen Signale und liefert seinerseits entsprechend Signale um eine anliegende Eingangsspannung vorzeichenrichtig mit einer Genauigkeit von 16 Bit zu erfassen. Bedingt durch die Synchronisation der Integrationsphasen mit der Netzfrequenz ergibt sich eine maximale Wandlungsrate von 31 Messungen pro Sekunde.

Dabei bestimmt das vom Rechner programmierbare Gainregister in welchem Spannungsbereich gearbeitet wird. Der Meßbereichsendwert ist dabei in 13 Stufen von ± 10 mV bis ± 50 V wählbar.

Mit dem kleinsten Endwert von ± 10 mV und der Isolationsspannung zwischen Primär- und Sekundärseite von ± 500 V ist es z.B. möglich, direkte Geberspannungen auf hohem Potential liegend zu messen.

Ein kompletter Softwaretreiber übernimmt die Grundinitialisierung und die Steuerung einer Messung, wozu auch eine Interrupt-Service-Routine gehört, welche das Ergebnis der Wandlung zur weiteren Verarbeitung in den Rechner holt. Liegen mehrere zu messende Analoggrößen vor, ist eine Erweiterung mit einer oder mehrerer Relaiskarten ECBS/R16 empfehlenswert.

Die Relaismultiplexerkarten erhalten von der Analogeingabe ECB/E1 zu Beginn der Abintegrationsphase Umschaltimpulse zur Umschaltung der Kanäle. Durch dieses Verfahren wird die maximale Wandlungsrate von 31 Messungen pro Sekunde nicht beeinträchtigt.



2. Adressierung


Die Einstellung der Basisadresse erfolgt an einem 5-fach-DIP-Schalter auf der Baugruppe.

Schalter-Adreßbit	Adressenform	Adreßbedarf
S1 - A7	5 Bit Basisadresse	8H
S2 - A6	3 Bit Arbeitsadressen	
S3 - A5		
S4 - A4		
S5 - A3		

mögliche Basisadressen:


Einstellbeispiel

dez.: 32, 40 240, 248
hex.: 20, 28 F0, F8
bin.: 00100000 11111000



2.1 Zuordnung der Steueradressen

A2	A1	A0	! Funktion
0	0	0	! STC Am9513 - Data Word
0	0	1	! STC AM9513 - Control Word
0	1	0	! Read Status
0	1	0	! Set Gain
0	1	1	! Start Impuls
1	0	0	! Z80A-CTC Kanal 0
1	0	1	! Z80A-CTC Kanal 1
1	1	0	! Z80A-CTC Kanal 2
1	1	1	! Z80A-CTC Kanal 3



3. Betriebsarten und ihre Programmierung

Die Baugruppe ECB/E1 wird in Ihren möglichen Betriebsarten durch Treibermodule softwaremäßig unterstützt.

Zum besseren Verständnis der nachfolgend genannten Betriebsarten sollen hier einige Begriffe erläutert werden:

Abtaste:	Meßzeit für das Einziehen eines Meßwertes auf einem Kanal.
Meßzyklusdauer:	Zeitspanne für das Einziehen eines Meßwertes für n-Kanäle.
Meßintervalldauer:	Zeitspanne zwischen dem zeitlichen Beginn zweier aufeinander folgender Meßzyklen.

3.1 Mögliche Benutzung der Baugruppe als Einkanalmeßsystem


Bei der Benutzung der Baugruppe als Einkanalmeßsystem sind zwei unterschiedliche Zeittaktsteuerungen realisierbar.

3.1.1 Einkanalige Messungen, die unmittelbar aufeinanderfolgen

In diesem Modus ergeben sich folgende Werte:

Abtaste/Kanal: ca. 33 ms
Meßzyklusdauer: ist hier gleich Abtaste
Meßintervalldauer: entfällt, da nur einkanalige Messung

Für diese Betriebsart sind die Treibermodule INIE1, SE1 und GE1 zu verwenden. Ein Programmierbeispiel für diese Anwendung zeigt das Fortran-Programm "E1T" auf, dessen Listing sich im Anhang befindet.



3.1.2 Einkanalige Messungen mit beliebig wählbarem Meßintervall

Um die Baugruppe ECB/E1 in diesem Modus einsetzen zu können, ist **zusätzlich** die Zähler-/Zeitgeberbaugruppe **ECB/TC** notwendig. Für diesen Modus ergeben sich dann folgende Werte:

Abtastrate/Kanal: ca. 33 ms
Meßzyklusdauer: entfällt, da nur einkanalige Messung
Meßintervalldauer: frei einstellbar, aber größer als 33 ms

Für diese Betriebsart sind die Treibermodule INIE1, SE1 und GE1 zu verwenden und zumindest das Modul SE1 in der Interrupt-Serviceroutine für die ECB/TC aufzurufen!

3.2 Mögliche Benutzung der Baugruppe ECB/E1 als Mehrkanalsystem

Diese Betriebsart setzt in jedem Fall den zusätzlichen Einsatz der Relaismultiplexer-Baugruppe ECBS/R16 voraus.

Die Relaismultiplexer-Baugruppe ECBS/R16 wird mit der ECB/E1 über Stecker ST2 verbunden.

3.2.1 Mehrkanalige Messung mit unmittelbar aufeinanderfolgenden Meßzyklen

Für diesen Modus ergeben sich folgende Werte:

Abtastrate/Kanal: ca. 33 ms
Meßzyklusdauer: Anzahl der Kanäle x 33 ms
Meßintervalldauer: ist hier gleich Meßzyklusdauer

Für diese Betriebsart sind die Treibermodule INIE1, SRE1 und GE1 zu verwenden.

Ein Programmierbeispiel für diese Ansteuerung zeigt das Fortran-Programm "E1TR", dessen Listing sich im Anhang befindet.



3.2.2 Mehrkanalige Messungen mit beliebig wählbarem Meßintervall


Diese Betriebsart setzt außer der Relaismultiplexerkarte ECBS/R16 auch noch die Zähler-/Zeitgeberbaugruppe ECB/TC voraus.

Es ergeben sich für diesen Modus folgende Werte:

Abtastrate/Kanal:	ca. 33 ms
Meßzyklusdauer:	Anzahl der Kanäle x 33 ms
Meßintervalldauer:	ist hier frei wählbar, mindestens aber Anzahl der Kanäle x 33 ms

Für diese Betriebsart sind die Treibermodule INIE1, SRE1 und GE1 zu verwenden und zumindest das Modul SRE1 in der Interrupt-Serviceroutine für die ECB/TC zu verwenden.

Hinweis: Die Weiterverarbeitung der Meßwerte ist erst nach Abschluß der gesamten Messung möglich.



3.3 Beispielprogramme

Zum Funktionstest sowie zum Abgleich der Analogeingabe-Baugruppe ECB/E1 können die Fortran-Beispielprogramme, deren Listing sich im Anhang befindet, herangezogen werden. Diese Programme erheben jedoch keinen Anspruch auf Funktionalität.

Ebenso befindet sich im Anhang ein Listing der einzelnen Assembler-Routinen der Treiberprogramme.

Die Funktion der einzelnen Beispielprogramme ist wie folgt:

Programme	Funktion
E1T	Kontinuierliche Messung mit ECB/E1 und Anzeige auf dem Bildschirm mit Voreinstellbarkeit des Eingangsbereiches.
E1TR	Kontinuierliche Messung und Darstellung von bis zu 16 Kanälen mit Relais-Mux ECBS/R16 mit Vorwahlmöglichkeit des Eingangsbereiches.


4. Abgleichvorschrift

Zum Abgleich der Platine werden folgende Geräte/Teile benötigt:

- Extenderplatine ECB/Y
- Eichquelle mit 0,006% Genauigkeit
- Trimmerschraubendreher
- Testprogramm E1T

Nullabgleich: Testprogramm E1T mit +/-10 mV Bereich Eingang kurzschließen mit P2 auf 0 V abgleichen.

Rangeabgleich: Testprogramm E1T mit Bereichsanwahl in dem am genauesten gemessen werden soll. Z.B. 1 V - Bereich einstellen mit P1 auf Endwert abgleichen.



! Vorhandene Eingangsbereiche ! Abweichung ! Genauigkeit !							
!	1	+/-	10 mV	!	9 uV	!	11 Bit
!	2	+/-	20 mV	!	9 uV	!	12 Bit
!	3	+/-	50 mV	!	11 uV	!	13 Bit
!	4	+/-	100 mV	!	15 uV	!	14 Bit
!	5	+/-	200 mV	!	18 uV	!	> 14 Bit
!	6	+/-	1 V	!	92 uV	!	> 14 Bit
!	7	+/-	2 V	!	122 uV	!	> 14 Bit
!	8	+/-	2,5 V	!	229 uV	!	> 14 Bit
!	9	+/-	5 V	!	458 uV	!	> 14 Bit
!	10	+/-	10 V	!	916 uV	!	> 14 Bit
!	11	+/-	20 V	!	1,8 mV	!	> 14 Bit
!	12	+/-	25 V	!	2,3 mV	!	> 14 Bit
!	13	+/-	50 V	!	4,6 mV	!	> 14 Bit


5. Steckerbelegung

ST2:

PIN 1, 3, 5, 7, 9	CASE (Gehäuseabschirmung des ext. Gerätes)
PIN 2	SIG.RET (Masse des Meßsignals)
PIN 4	IN (analoger Eingang)
PIN 6	verbunden mit ST3 PIN 1, 2, 3, 4, 5
PIN 8	verbunden mit ST3 PIN 6, 7, 8, 9, 10
PIN 10	N.C.

Das zu messende Signal liegt zwischen IN und SIG.RET an.


ST3: siehe ST2 PIN 6 und PIN 8




6. Anschlußbeispiele


a) Verbindung von Signalquelle zur ECB/E1 mit Koaxkabel

Der Schutzschirm der ECB/E1 wird mit dem Schaltungsnullpunkt verbunden. Dazu ist es notwendig den Lötjumper auf der Unterseite der ECB/E1-Baugruppe zwischen Pin 1 und Pin 2 von ST2 zu schließen.



b) Verbindung von Signalquelle zur ECB/E1 durch Kabel mit zwei Innenleitern





7. Beschreibung der Treibermodule zur Analogeingabe-Baugruppe ECB/E1

Diese Treiber sind so aufgebaut, daß ein Aufruf aus Fortran oder Microsoft-Basic möglich ist.

Die Übergabeparameter sind in Fortran als INTEGER*2 deklariert (standardmäßig bei Fortran und Microsoft-Basic falls keine andere Deklaration erfolgt).

7.1 Initialisierung der Analogeingabe-Baugruppe ECB/E1 und Programmierung des Spannungsbereiches

Bevor mit der Analogeingabe-Baugruppe ECB/E1 gearbeitet werden kann, ist es notwendig die Karte zu initialisieren. Der Aufruf des Initialisierungsmoduls aus Fortran und Microsoft-Basic lautet:

CALL INIE1(CAS,GAIN)

Parameter:

CAS Kartenadresse der ECB/E1
 mögliche Adressen:
 Dezimal: 32, 40,, 240, 248
 Hexadezimal: 20, 28,, F0, F8


GAIN Verstärkungsfaktor
 Mögliche Werte: 1, 2,, 12, 13

GAIN = 1: +/- 10 mV	GAIN = 8: +/- 2,4 V
= 2: +/- 20 mV	= 9: +/- 5 V
= 3: +/- 50 mV	=10: +/- 10 V
= 4: +/- 100 mV	=11: +/- 20 V
= 5: +/- 200 mV	=12: +/- 25 V
= 6: +/- 1 V	=13: +/- 50 V
= 7: +/- 2 V	

Der Aufruf der Initialisierungsroutine aus Assembler heraus lautet:

```
      .  
      .  
      .  
LD     HL,CAS  
LD     DE,GAIN  
CALL  INIE1  
      .  
      .  
      .  
      .
```

CAS:	DEFW 80H	;Baugruppenadresse 128
GAIN:	DEFW 0AH	;Verstärkung +/- 10 V



Da der auf der Baugruppe befindliche Z80A-CTC durch das Modul INIE1 auf Interrupt initialisiert wird, ist darauf zu achten, daß in der Interrupttabelle ab Adresse 80H Platz für zwei Einträge (4 Byte) vor der Initialisierung frei ist, da dies sonst zu Systemkonflikten führen kann.

7.2 Starten der Messung

Das Starten einer Messung erfolgt durch den Aufruf des Treibermoduls SE1. Der Aufruf aus Fortran und Micorsoft-Basic lautet:

CALL SE1 (CAS,VALUE,N)

Parameter:

Eingabe: CAS	Kartenadresse der ECB/E1 wie bei Kapitel 7.1
N	Anzahl der durchzuführenden Messungen (1, 2,)
Ausgabe: VALUE	Feld mit den eingezogenen Meßwerten; die Meßwerte sind INTEGER*2 Größen bzw. 2 Byte groß. Das Feld VALUE muß mindestens 2*N groß dimensioniert sein.

Im ersten Byte bzw. INTEGER*2 Element ist der Digitalwert enthalten, im zweiten das Vorzeichen und der Status (Overflow).


Die Spannung und der Status des K-ten Meßwertes ergeben sich wie folgt:

$SPANNUNG = VALUE (2 * K + 1) / 32768$

IF BIT0 OF VALUE (2 * K + 2) = 'L' THEN NEGATIVE VOLTAGE

IF BIT1 OF VALUE (2 * K + 2) = 'L' THEN OVERFLOW

(0 ≤ K ≤ N-1)



Der Aufruf dieser Routine aus Assembler heraus lautet:

```
.  
. .  
LD HL,CAS  
LD DE,VALUE  
LD BC,N  
CALL SE1  
. .  
.
```

CAS: DEFW 80H ;Baugruppenadresse 128
VALUE: DEFS 20H ;32 Speicherplätze für die
 ;Meßwerte
N: DEFW 10H ;16 Messungen

7.3 Starten der Messung mit der Relaismultiplexer-Baugruppe ECBS/R16

Eine Messung, bei der mehrere Kanäle gemultiplext werden, kann durch den Aufruf des Treibermoduls SRE1 gestartet werden.


Der Aufruf dieses Moduls aus Fortran und Microsoft-Basic heraus lautet:

CALL SRE1(CAS,CASR,VALUE,M,N)

Parameter:

Eingabe: CAS	Kartenadresse der ECB/E1 wie bei 7.1
CASR	Kartenadresse der ECBS/R16 Mögliche Adressen: Dezimal: 00, 01, 02, 254, 255 Hexadezimal: 00, 01, 02, FE, FF
M	Anzahl der zu messenden Kanäle Mögliche Werte: 1, 2, 16
N	Anzahl der pro Kanal durchzuführenden Messungen (1, 2,,)
Ausgabe: VALUE	Feld mit den eingezogenen Meßwerten; die Meß- werte sind INTEGER*2 Größen bzw. 2 Byte groß. Das Feld VALUE muß mindestens $2 * N * M$ groß dimensioniert sein

Zur Bedeutung des Inhalts von VALUE siehe 7.2.



Der Aufruf dieser Routine aus Assembler heraus lautet:

```
      .  
      .  
      .  
LD     HL,CAS  
LD     DE,CASR  
LD     BC,BCPOI  
CALL   SRE1  
      .  
      .  
      .  
  
CAS:    DEFW 80H                ;Baugruppenadresse  
                                ;ECB/E1 = 128  
  
CASR:    DEFW 7FH              ;Baugruppenadresse  
                                ;ECBS/R16 = 127  
  
BCPOI:   DEFW VALUE,M,N  
  
      VALUE: DEFS AOH          ;160 Speicherplätze  
            M: DEFW 10H        ;16 Kanäle  
            N: DEFW 05H        ;5 Messungen pro Kanal
```

7.4 Überprüfung auf Ende der Messung; Warten auf Messungsende

Nach dem Start einer Meßreihe besteht die Möglichkeit, den Rechner mit anderen Aufgaben auszulasten. Wenn jedoch die Meßwerte zur Weiterverarbeitung benötigt werden, so ist es notwendig zu überprüfen, ob die Meßreihe abgeschlossen ist und gegebenenfalls auf deren Ende zu warten. Dies erfolgt durch den Aufruf des Treibermoduls GE1.


Der Aufruf aus Fortran und Microsoft-Basic lautet:

CALL GE1(CAS)

Zur Bedeutung des Eingabeparameters "CAS" siehe Kapitel 7.1!

Aus Assembler heraus lautet der Aufruf dieser Routine:

```
      .  
      .  
      .  
LD     HL,CAS  
CALL   GE1  
      .  
      .  
      .  
CAS    DEFW 080H                ;Baugruppenadresse 128
```




7.5 Schema für die Parameterübergabe

Die Übergabe der Parameter erfolgt nach folgendem Schema:

- a) Bei Übergabe von weniger als vier Parametern:
Adresse des ersten Parameter in HL
Adresse des zweiten Parameter in DE
Adresse des dritten Parameter in BC
- b) Bei Übergabe von mehr als 3 Parametern:
Adresse des ersten Parameter in HL
Adresse des dritten Parameter in DE
Dritter Parameter unter der Adresse die in BC und BC+1 steht.
Vierter Parameter unter der Adresse die in BC+2 und BC+3 steht, usw...

8. Technische Daten

Spannungsversorgung:	+5V, +15V, +/-5%
Stromaufnahme:	600mA, 70mA
Analogmodul-Auflösung:	16 Bit
Linearität:	0,006%
CMRR:	150 dB
CMV:	1000 V
Meßbereiche:	+/- 10, 20, 50, 100, 200 mV +/- 1, 2, 5, 5, 10, 20, 25, 50 V
Eingangswiderstand:	FSR <= +/- 5V, 10 E8 Ohm min. FSR => +/- 5V, 10 E6 Ohm nom.
Insolationsspannung:	+/- 500 VDC oder +/- 500 VAC (Vss)
Wandlungsrate:	31 Messungen pro Sekunde synchronisiert mit 50 Hz




9. Besonderheiten der Baugruppe


Die Baugruppe besitzt zwei separate Eingänge für den Anschluß einer externen Konstantstrom-/Spannungsquelle zur Geberspeisung über die Relaismultiplexerbaugruppe. .pa

Funktionsweise

Die Relais der ECBS/R16 sind vierpolig durchschaltend. Zwei Pole werden für das Durchschalten der Meßsignale zur ECB/E1 benutzt. Die restlichen zwei Pole schalten jeweils den Speisestrom oder die Speisespannung des angewählten Kanales von der ECB/E1 durch.


Es können generell Geber vom Typ PT100 sowie DMS-Brücken versorgt werden. Der Ein-/Ausgangsstecker ist für diese Anschlußtechnik ausgelegt.





10. Beispielprogramme

```
C
C  BEISPIELPROGRAMM FÜR DIE ANALOGEINGABEBAUGRUPPE ECB/E1
C      (EINKANALIGE MESSWERTERFASSUNG)
C
C      23.08.1984 by H.W.
C
C      INTEGER CAS,CASR,M,N,VALUE(1000),GAIN,CUP,INV,CSW
C      INTEGER*1 TEXT(2)
C      DOUBLE PRECISION ANALOG
C      REAL FAKTOR(13)
C      DATA FAKTOR /.01,.02,.05,.1,.2,1.,2.,2.5,5.,10.,20.,25.,50./
C
C  MONITOR-STEUERZEICHEN
C
C  CURSOR UP
C      CUP=26
C  INVERT ON/OFF
C      INV=17
C  CURSOR ON/OFF
C      CSW=20
C
C  PARAMETER EINLESEN ÜBER BILDSCHIRM
C
C      WRITE(1,1)CSW
C      1  FORMAT(1H ,A1)
C      DO 5 I=1,1000
C      VALUE(I)=255
C      5  CONTINUE
C      WRITE(1,10)
C      10  FORMAT(//,3H , 'BITTE BAUGRUPPENADRESSE DER ECB/E1 EINGEBEN:')
C      READ(1,20)CAS
C      20  FORMAT(I3)
C      25  WRITE(1,26)
C      26  FORMAT(//,1H , '(01) +- 10 mVOLT (02) +- 20 mVOLT',/,
C      * 1H , '(03) +- 50 mVOLT (04) +-100 mVOLT',/,
C      * 1H , '(05) +-200 mVOLT (06) +- 1.0 VOLT',/,
C      * 1H , '(07) +- 2.0 VOLT (08) +- 2.5 VOLT',/,
C      * 1H , '(09) +- 5.0 VOLT (10) +-10.0 VOLT',/,
C      * 1H , '(11) +-20.0 VOLT (12) +-25.0 VOLT',/,
C      * 1H , '(13) +-50.0 VOLT',//)
C      WRITE(1,30)
C      30  FORMAT(//,3H , 'VERSTÄRKUNGSFAKTOR (1-13):')
C      READ(1,40)GAIN
C      40  FORMAT(I2)
C
C  INITIALISIERUNG DER ECB/E1
C
C      CALL INIE1(CAS,GAIN)
C      WRITE(1,50)
C      50  FORMAT(//,3H , 'ANZAHL DER MESSUNGEN:')
C      READ(1,60)N
C
C
C  STARTEN DER MESSREIHE
```




```
C
DO 123 L=1,N
60  FORMAT(I5)
    CALL SE1(CAS,VALUE,1)
    CALL GE1(CAS)
    ANALOG=FLOAT(VALUE(1))*1.
    STATUS=MOD(VALUE(2),4)


C
C  BERECHNUNG DES ANALOGWERTES
C
    ANALOG=ANALOG*FAKTOR(GAIN)/32768
    IF ((STATUS.EQ.1).OR.(STATUS.EQ.3))ANALOG=-ANALOG
    TEXT(1)=' '
    IF (STATUS.GT.1) TEXT(1)='0'

C
C  MESSWERTAUSGABE
C
    IF (STATUS.LE.1) WRITE (1,111)INV,TEXT(1),ANALOG,INV,L,CUP
    IF (STATUS.GT.1) WRITE (1,112)INV,INV,L,CUP
123  CONTINUE

C
C  ABFRAGE OB NEUE MESSREIHE GESTARTET WERDEN SOLL
C
    WRITE(1,115)
115  FORMAT(1H ,//,' NOCH EINE MESSREIHE DURCHFÜHREN (Y/N)?')
    READ(1,116)TEXT(2)
116  FORMAT(1A1)
    IF (TEXT(2).EQ.78) GOTO 888
    IF (TEXT(2).EQ.89) GOTO 25
    GOTO 123
888  CONTINUE
    WRITE(1,1)CSW
    STOP
111  FORMAT(1H ,A1,'SPANNUNG:',A1,F7.4,' VOLT',A1,' ',I6,A1)
112  FORMAT(1H ,A1,'UEBERLAUF ',A1,' ',I6,A1)
END
```



```
C
C
C   BEISPIELPROGRAMM FÜR DIE ECB/E1 IN VERBINDUNG MIT DER
C   MULTIPLEXERBAUGRUPPE ECBS/R16.
C
C   23.03.1984 by H.W.
C
C   INTEGER CAS,CASR,M,N,VALUE(1000),GAIN,CUP,INV,CSW,MUX
C   INTEGER*1 TEXT(2)
C   DOUBLE PRECISION ANALOG
C   REAL FAKTOR(13)
C   DATA FAKTOR /.01,.02,.05,.1,.2,1.,2.,2.5,5.,10.,20.,25.,50./
C
C   CONTROL-CODES
C
C   CURSOR UP
C   CUP=26
C   INVERT ON/OFF
C   INV=17
C   CURSOR ON/OFF
C   CSW=20
C
C   PARAMETER EINLESEN ÜBER BILDSCHIRM
C
C   WRITE(1,1)CSW
C   1  FORMAT(1H ,A1)
C   DO 5 I=1,1000
C   VALUE(I)=255
C   5  CONTINUE
C   WRITE(1,10)
C   10 FORMAT(//,3H , 'BITTE BAUGRUPPENADRESSE DER ECB/E1 EINGEBEN: ')
C   READ(1,20)CAS
C   20 FORMAT(I3)
C   WRITE(1,21)
C   21 FORMAT(/,3H , 'BITTE BAUGRUPPENADRESSE DER ECBS/R16 EINGEBEN: ')
C   READ(1,20)CASR
C   25 WRITE(1,26)
C   26 FORMAT(//,1H , '(01) +- 10 mVOLT   (02) +- 20 mVOLT' ,/,
C   * 1H , '(03) +- 50 mVOLT   (04) +-100 mVOLT' ,/,
C   * 1H , '(05) +-200 mVOLT   (06) +- 1.0 VOLT' ,/,
C   * 1H , '(07) +- 2.0 VOLT   (08) +- 2.5 VOLT' ,/,
C   * 1H , '(09) +- 5.0 VOLT   (10) +-10.0 VOLT' ,/,
C   * 1H , '(11) +-20.0 VOLT   (12) +-25.0 VOLT' ,/,
C   * 1H , '(13) +-50.0 VOLT' ,//)
C   WRITE(1,30)
C   30 FORMAT(//,3H , 'VERSTÄRKUNGSFAKTOR (1-13): ')
C   READ(1,40)GAIN
C   40 FORMAT(I2)
C
C   INITIALISIERUNG DER ECB/E1
```


```
C
    CALL INIE1(CAS,GAIN)
    WRITE(1,50)
50  FORMAT(//,3H  , 'ANZAHL DER MESSUNGEN PRO KANAL:')
    READ(1,60)N
60  FORMAT(I5)
    WRITE(1,61)
61  FORMAT(//,3H  , 'ANZAHL DER MULTIPLEX-KANAELE:')
    READ(1,20)MUX

C
C  STARTEN DER MESSUNG
C
    DO 123 L=1,N
    CALL SRE1(CAS,CASR,VALUE,MUX,1)
    CALL GE1(CAS)
    IMUX=(2*MUX)
    DO 77 IML=1,IMUX,2
    ANALOG=FLOAT(VALUE(IML+0))*1.
    STATUS=MOD(VALUE(IML+1),4)


C
C  BERECHNUNG DES ANALOGWERTES
C
    ANALOG=ANALOG*FAKTOR(GAIN)/32768
    IF ((STATUS.EQ.1).OR.(STATUS.EQ.3))ANALOG=-ANALOG
    TEXT(1)=' '
    IF (STATUS.GT.1) TEXT(1)='O'

C
C  MESSWERTAUSGABE
C
    KANAL=(IML+1)/2
    IF (STATUS.LE.1) WRITE (1,111)INV,KANAL,TEXT(1),ANALOG,INV
    IF (STATUS.GT.1) WRITE (1,112)INV,KANAL,INV
77  CONTINUE
    IF (L.EQ.(N-0)) GOTO 123
    DO 155 IM=1,MUX
    WRITE(1,113)CUP,CUP
155  CONTINUE
123  CONTINUE

C
C  ABFRAGE OB PROGRAMM WIEDER GESTARTET WERDEN SOLL
C
    WRITE(1,115)
115  FORMAT(1H ,//, ' NOCH EINE MESSREIHE DURCHFÜHREN (Y/N)?')
    READ(1,116)TEXT(2)
116  FORMAT(1A1)
    IF (TEXT(2).EQ.78) GOTO 888
    IF (TEXT(2).EQ.89) GOTO 25
    GOTO 123
888  CONTINUE
    WRITE(1,1)CSW
    STOP
```




```
111 FORMAT(1H ,A1,'KANAL:',I2,'      SPANNUNG:'A1,F7.4,' VOLT',A1)
112 FORMAT(1H ,A1,'KANAL:',I2,'      UEBERLAUF      ',A1)
113 FORMAT(1H ,2A1)
END
```

11. Treiberprogramm zur ECB/E1

```
;*****
;*TREIBERPROGRAMM FUER DIE KARTE ECB/E1 (UND ECBS/R16) *
;*****
;*AUTOR: 15.9.1981 / U. H. *
;*REVISION: 6.10.1982 / C. S. *
;*REVISION:24.08.1984 / H. W. *
;*****
;*AUFRUFE: *
;* - CALL INIE1 (CASE,GAIN) *
;* Initialisierung der Karte ECB/E1 *
;* - CALL SE1 (CASE,VALUE,N) *
;* Start der Messung *
;* - CALL SRE1 (CASE,CASR,VALUE,M,N) *
;* Start der Messung mit ECBS/R16 und ECB/E1 *
;* - CALL GE1 (CASE) *
;* Warten auf Ende der Messung *
;*PARAMETER: *
;* - CASE (INTEGER*2) KARTENADRESSE ECB/E1 *
;* - CASR (INTEGER*2) KARTENADRESSE ECBS/R16 *
;* - GAIN (INTEGER*2) VERSTAERKUNGSFAKTOR *
;* 1 +-10 MVOLT *
;* 2 +-20 MVOLT *
;* 3 +-50 MVOLT *
;* 4 +-100 MVOLT *
;* 5 +-200 MVOLT *
;* 6 +-1.0 VOLT *
;* 7 +-2.0 VOLT *
;* 8 +-2.5 VOLT *
;* 9 +-5.0 VOLT *
;* 10 +- 10 VOLT *
;* 11 +- 20 VOLT *
;* 12 +- 25 VOLT *
;* 13 +- 50 VOLT *
;* - WERT (INTEGER*2 FELD MIT 2*M*N ELEMENTEN) *
;* EINGELESENER WERT *
;* WERT(1) *
;* MOD(WERT(2),2).EQ.0 POSITIV *
;* MOD(WERT(2),2).NE.0 NEGATIV *
;* WERT(2).GT.1 UEBERLAUF *
;* - M (INTEGER*2) ANZAHL DER KANAEL PRO MESS. *
;* - N (INTEGER*2) ANZAHL DER MESS. PRO KANAL *
;*****
;
;TITLE $TE1
;
; .Z80
;ENTRY INIE1,SE1,GE1,SRE1
;
;PARAMETER
```



```
;
      INTTAB EQU 80H      ;EINTRAG IN DER INTERRUPTTABELLE
CASE: DEFS 1             ;Kartenadresse ECB/E1
CASR: DEFS 1             ;(Kartenadresse ECBS/R16)
GAIN: DEFS 1             ;Verstaerkungsfaktor
MX:   DEFS 1             ;(Kanalzahl)
MA:   DEFS 1             ;(aktuelle Kanalzahl)
N:    DEFS 2             ;Messungsanzahl
VALUE: DEFS 2            ;Adresse des Wertefeldes
COUNTR: DEFS 2           ;Adresse Feldelement
STATUS: DEFS 1           ;Interruptzustaende
STACK: DEFS 60           ;HILFSSTACK
STACKA:
SAVEST: DEFS 2           ;STACKADRESSE NORMALMODUS


;MACRODEFINITIONEN

PUSHR MACRO
      PUSH AF
      PUSH BC
      PUSH DE
      PUSH HL
      ENDM

POPR MACRO
      POP HL
      POP DE
      POP BC
      POP AF
      ENDM

;*****
;INITIALISIERUNG DER KARTE E1*
;*****

INIE1: PUSHR
      LD A,(HL)           ;Kartenadresse abspeichern
      LD (CASE),A
      LD A,(DE)           ;Verstaerkungsfaktor abspeichern
      LD (GAIN),A
      CALL CTC12          ;CTC initialisieren
      CALL STC            ;STC initialisieren
      CALL SGAIN          ;Verstaerkungsfaktor programmieren
      POPR
      RET
```



```
;*****  
;Programmierung des Verstaerkungsfaktors*  
;*****
```


```
GANTAB: DEFB 41H  
        DEFB 21H  
        DEFB 42H  
        DEFB 44H  
        DEFB 24H  
        DEFB 48H  
        DEFB 28H  
        DEFB 50H  
        DEFB 30H  
        DEFB 0C8H  
        DEFB 0A8H  
        DEFB 0DOH  
        DEFB 0BOH
```

```
SGAIN:  LD HL,GANTAB      ;Berechnung des Verstaerkungsfaktors  
        LD A,(GAIN)  
        LD E,A  
        LD D,OOH  
        ADD HL,DE  
        DEC HL  
        LD A,(CASE)      ;Adresse des Verstaerkungsregisters  
        INC A  
        INC A  
        LD C,A  
        LD A,(HL)  
        OUT (C),A        ;GAIN-REG. SETZEN  
        LD (GAIN),A
```

```
LABL:   LD HL,0800H      ;Warteschleife 12.8 Msec  
        DEC HL          ;1.5 uSec  
        LD A,H          ;1.0 uSec  
        OR L            ;1.0 uSec  
        CP O            ;1.0 uSec  
        JR NZ,LABL      ;1.75uSec Summe 6.25 uSec  
        RET
```

```
;*****  
;CTC initialisieren*  
;*****
```

```
CTC12:  LD A,(CASE)      ;CTC Adresse  
        INC A  
        INC A  
        INC A  
        INC A  
        LD C,A  
        LD D,INTTAB      ;Interrupt Vektor  
        LD E,0D7H        ;Steuerwort CTC  
        LD B,001H        ;Zaehlkonstante
```

```
OUT    (C),E           ;Kanal 0
OUT    (C),B
OUT    (C),D
INC     C               ;Kanal 1
OUT    (C),E
OUT    (C),B
OUT    (C),D
LD      L,D             ;Interrupt Adressen in Tabelle
LD      A,I
LD      H,A
LD      BC,IN1          ;INTERRUPTVECTOR 1
LD      (HL),C
INC     HL
LD      (HL),B
INC     HL
LD      BC,IN2          ;INTERRUPTVECTOR 2
LD      (HL),C
INC     HL
LD      (HL),B
RET
```

```
;*****
;STC initialisieren*
;*****
```


```
TABEL:  DEFB    OFFH           ;MASTER RESET
        DEFB    OE7H           ;ENTER 8-BIT MODE
        DEFB    OE6H           ;GATE ON FOUT
        DEFB    OE0H           ;ENABLE DP-SEQUENCING
        DEFB    OE5H           ;CLEAR OUTPUT 5
        DEFB    01H            ;DP AUF COUNTER 1 =>CMR

; COUNTER 1 - ZAEHLT IMPULSE VON F1=6,5536MHZ WENN G1 UND G1A "HIGH"
        DEFW    8108H           ;F1 QUELLE
        DEFW    0000H           ;ANFANGSWERT
        DEFW    0000H

; COUNTER 2 - MODE I
        DEFW    OC241H          ;CMR
        DEFW    1A00H           ;LOAD-REG. WERT FUER 2MSEC
        DEFW    8000H           ;HOLD-REG. " " 20MSEC-10USEC

; COUNTER 3 - MODE C
        DEFW    OC301H          ;CMR
        DEFW    OFFFFH         ;LOAD-REG.WERT FUER 20MSEC
        DEFW    0000H

; COUNTER 4 - MODE C
        DEFW    OC401H          ;CMR
        DEFW    8010H           ;LOAD-REG. WERT FUER 10MSEC
        DEFW    0000H
```



```
; COUNTER 5 - MODE C
      DEFW    OCBO2H      ;CMR
      DEFW    0010H      ;LOAD-REG. WERT FUER 1,2 USEC
      DEFW    0000H
```

```
; MASTER MODE REG.
      DEFB    17H        ;MMR
      DEFW    4200H      ;F1:10
```


```
STC:  LD      A,(CASE)   ;PROGRAMMIERUNG STC
      INC     A
      LD      C,A
      LD      HL,TABEL
      LD      B,06H
      OTIR
      DEC     C          ;DATEN ADR. STC
      LD      B,30D
      OTIR
      INC     C
      LD      B,01H
      OTIR
      DEC     C
      LD      B,02H
      OTIR
      RET
```

```
;*****
;Start der ECB/E1*
;*****
```

```
SE1:  PUSHR
      CALL    PAR1      ;Abspeicherung Parameter
      CALL    START    ;Messung starten
      POPR
      RET
```

```
;*****
;Start ECBS/R16 und ECB/E1*
;*****
```

```
SRE1: PUSHR
      CALL    PAR2      ;Abspeicherung Parameter
      CALL    START    ;Messung starten
      POPR
      RET
```



```
;*****  
;Pruefung, ob Messung beendet*  
;*****
```


```
GE1:   PUSH    AF  
GE11:  LD      A,(STATUS)  
       CP      1  
       JR      NZ,GE11  
       POP     AF  
       RET
```

```
;*****  
;Abspeicherung der Parameter ECB/E1*  
;*****
```

```
PAR1:  LD      A,0           ;Vorbesetzung STATUS  
       LD      (STATUS),A  
       LD      (CASR),A  
       LD      A,(HL)       ;Kartenadresse ECB/E1  
       LD      (CASE),A  
       LD      (VALUE),DE   ;Adresse Parameterblock  
       LD      A,(BC)       ;Anzahl der Messungen  
       LD      (N),A  
       INC     BC  
       LD      A,(BC)  
       LD      (N+1),A  
       RET
```

```
;*****  
;Abspeicherung der Parameter ECBS/R16 - ECB/E1*  
;*****
```

```
PAR2:  LD      A,0           ;Vorbesetzung STATUS  
       LD      (STATUS),A  
       LD      (MA),A       ;Vorbesetzung kanalzahl  
       LD      A,(HL)       ;Kartenadresse ECB/E1  
       LD      (CASE),A  
       LD      A,(DE)       ;Kartenadresse ECBS/R1  
       LD      (CASR),A  
       LD      H,B  
       LD      L,C  
       LD      C,A         ;Ansteuerung ECBS/R16  
       LD      A,10H  
       OUT     (C),A  
       LD      A,(HL)       ;Wertefeld Adresse  
       LD      (VALUE),A  
       INC     HL  
       LD      A,(HL)  
       LD      (VALUE+1),A  
       INC     HL  
       LD      E,(HL)       ;Kanalzahl
```

```

        INC     HL
        LD      D,(HL)
        LD      A,(DE)
        LD      (MX),A
        INC     HL
        LD      E,(HL)
        INC     HL
        LD      D,(HL)
        LD      A,(DE)
        LD      (N),A
        INC     DE
        LD      A,(DE)
        LD      (N+1),A


LABL2:   LD      HL,0800H           ;Warteschleife 12.8 Msec
        DEC     HL               ;1.5 uSec
        LD      A,H              ;1.0 uSec
        OR      L                ;1.0 uSec
        CP      0               ;1.0 uSec
        JR      NZ,LABL2         ;1.75uSec Summe 6.25 uSec
        RET

;*****
;Start der Messung*
;*****

TABEL2: DEFB    OE5H             ;CLEAR OUTPUT 5
        DEFB    41H             ;COUNTER1 NULL SETZEN
        DEFB    3FH             ;ALLE COUNTER FREIGEBEN
START:  PUSH    HL              ;Adresse Startregister
        LD      A,(CASE)
        INC     A
        LD      C,A
        LD      B,03H
        LD      HL,TABEL2
        OTIR                    ;STC VORBEREITEN UND FREIGEBEN
        INC     C
        INC     C               ;C ==> F3H-STARTIMPULS
        LD      A,(GAIN)
        OUT     (C),A           ;SENDEN
        POP     HL
        RET

;*****
;Ergebnis abspeichern*
;*****

SAVE:   LD      HL,(VALUE)
        LD      A,(CASE)
        INC     A
```




```
LD      C,A                                ;CONTROL-ADR.STC
LD      A,0A1H                            ;COUNTER 1 IN HOLD-
OUT     (C),A                             ;REG. RETTEN
LD      A,11H                             ;DP AUF HOLD-REG.
OUT     (C),A
DEC     C                                ;DATEN-ADR. STC
IN      A,(C)                             ;LOW-BYTE
LD      (HL),A
INC     HL
IN      A,(C)                             ;HIGH-BYTE
LD      (HL),A
INC     HL
INC     C
INC     C                                ;C ==> F2H-STATUSREG. ADR.
IN      A,(C)
LD      (HL),A
INC     HL
LD      A,0
LD      (HL),A
INC     HL
LD      (VALUE),HL
RET

;*****
; INTERRUPT ROUTINES*
;*****

IN1:    LD      (SAVEST),SP                ;Spezial Stack
LD      SP,STACKA
PUSHR
CALL    SAVE                            ;Absicherung der Werte
LD      A,(CASR)                         ;Mit Relaiskarte?
CP      0
JR      NZ,IN12
LD      HL,(N)                           ;Alle Messungen / Kanal?
DEC     HL
LD      (N),HL
LD      A,H
OR      L
CP      0
JR      NZ,IN13
LD      A,1                             ;Ende der Messreihe
LD      (STATUS),A
JP      IN1E

IN12:   LD      A,(MX)                    ;Relaisfolge beendet?
LD      B,A
LD      A,(MA)
CP      B
JR      NZ,IN13
```





```
LD      A, OFFH      ;Relais ruecksetzen
LD      (MA), A
LD      HL, (N)       ;Alle Messungen / Kanal?
DEC     HL
LD      (N), HL
LD      A, H
OR      L
CP      O
JR      NZ, IN13
LD      A, 1          ;Ende der Messreihe
LD      (STATUS), A
LD      A, (CASR)     ;Abschalten Relais ECBS/R16
LD      C, A
LD      A, OH
OUT     (C), A
JP      IN1E
IN13:   CALL START    ;Start weiterer Messungen
IN1E:   POPR
LD      SP, (SAVEST)
EI
RETI


IN2:    LD      (SAVEST), SP      ;Spezial Stack
LD      SP, STACKA
PUSH    AF
LD      A, (CASR)               ;Mit Relaiskarte?
CP      O
JR      Z, IN21
LD      A, (CASR)               ;Durchschalten naechstes Relais
LD      C, A
LD      A, (MA)
INC     A
LD      (MA), A
OR      1OH
OUT     (C), A
IN21:   POP     AF
LD      SP, (SAVEST)
EI
RETI

IN3:    LD      (SAVEST), SP      ;Externe Triggerung
LD      SP, STACKA
PUSH    AF
LD      A, 4
LD      (STATUS), A
POP     AF
LD      SP, (SAVEST)
EI
RETI

FIN:    END
```



160MM X 100MM



BESTUECKUNGSPLAN

P380 1.2M2 AN/E1 F.KAP